creation_sourcecode: // SPDX-License-Identifier: MIT
  // File: @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol
  
  
  // OpenZeppelin Contracts (last updated v5.0.0) (token/ERC721/IERC721Receiver.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @title ERC721 token receiver interface
   * @dev Interface for any contract that wants to support safeTransfers
   * from ERC721 asset contracts.
   */
  interface IERC721Receiver {
      /**
       * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
       * by `operator` from `from`, this function is called.
       *
       * It must return its Solidity selector to confirm the token transfer.
       * If any other value is returned or the interface is not implemented by the recipient, the transfer will be
       * reverted.
       *
       * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
       */
      function onERC721Received(
          address operator,
          address from,
          uint256 tokenId,
          bytes calldata data
      ) external returns (bytes4);
  }
  
  // File: @openzeppelin/contracts/token/ERC721/utils/ERC721Holder.sol
  
  
  // OpenZeppelin Contracts (last updated v5.0.0) (token/ERC721/utils/ERC721Holder.sol)
  
  pragma solidity ^0.8.20;
  
  
  /**
   * @dev Implementation of the {IERC721Receiver} interface.
   *
   * Accepts all token transfers.
   * Make sure the contract is able to use its token with {IERC721-safeTransferFrom}, {IERC721-approve} or
   * {IERC721-setApprovalForAll}.
   */
  abstract contract ERC721Holder is IERC721Receiver {
      /**
       * @dev See {IERC721Receiver-onERC721Received}.
       *
       * Always returns `IERC721Receiver.onERC721Received.selector`.
       */
      function onERC721Received(address, address, uint256, bytes memory) public virtual returns (bytes4) {
          return this.onERC721Received.selector;
      }
  }
  
  // File: @openzeppelin/contracts/utils/Context.sol
  
  
  // OpenZeppelin Contracts v4.4.1 (utils/Context.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Provides information about the current execution context, including the
   * sender of the transaction and its data. While these are generally available
   * via msg.sender and msg.data, they should not be accessed in such a direct
   * manner, since when dealing with meta-transactions the account sending and
   * paying for execution may not be the actual sender (as far as an application
   * is concerned).
   *
   * This contract is only required for intermediate, library-like contracts.
   */
  abstract contract Context {
      function _msgSender() internal view virtual returns (address) {
          return msg.sender;
      }
  
      function _msgData() internal view virtual returns (bytes calldata) {
          return msg.data;
      }
  }
  
  // File: @openzeppelin/contracts/access/Ownable.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)
  
  pragma solidity ^0.8.0;
  
  
  /**
   * @dev Contract module which provides a basic access control mechanism, where
   * there is an account (an owner) that can be granted exclusive access to
   * specific functions.
   *
   * By default, the owner account will be the one that deploys the contract. This
   * can later be changed with {transferOwnership}.
   *
   * This module is used through inheritance. It will make available the modifier
   * `onlyOwner`, which can be applied to your functions to restrict their use to
   * the owner.
   */
  abstract contract Ownable is Context {
      address private _owner;
  
      event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);
  
      /**
       * @dev Initializes the contract setting the deployer as the initial owner.
       */
      constructor() {
          _transferOwnership(_msgSender());
      }
  
      /**
       * @dev Throws if called by any account other than the owner.
       */
      modifier onlyOwner() {
          _checkOwner();
          _;
      }
  
      /**
       * @dev Returns the address of the current owner.
       */
      function owner() public view virtual returns (address) {
          return _owner;
      }
  
      /**
       * @dev Throws if the sender is not the owner.
       */
      function _checkOwner() internal view virtual {
          require(owner() == _msgSender(), "Ownable: caller is not the owner");
      }
  
      /**
       * @dev Leaves the contract without owner. It will not be possible to call
       * `onlyOwner` functions. Can only be called by the current owner.
       *
       * NOTE: Renouncing ownership will leave the contract without an owner,
       * thereby disabling any functionality that is only available to the owner.
       */
      function renounceOwnership() public virtual onlyOwner {
          _transferOwnership(address(0));
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`).
       * Can only be called by the current owner.
       */
      function transferOwnership(address newOwner) public virtual onlyOwner {
          require(newOwner != address(0), "Ownable: new owner is the zero address");
          _transferOwnership(newOwner);
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`).
       * Internal function without access restriction.
       */
      function _transferOwnership(address newOwner) internal virtual {
          address oldOwner = _owner;
          _owner = newOwner;
          emit OwnershipTransferred(oldOwner, newOwner);
      }
  }
  
  // File: @openzeppelin/contracts/utils/math/SignedMath.sol
  
  
  // OpenZeppelin Contracts (last updated v4.8.0) (utils/math/SignedMath.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Standard signed math utilities missing in the Solidity language.
   */
  library SignedMath {
      /**
       * @dev Returns the largest of two signed numbers.
       */
      function max(int256 a, int256 b) internal pure returns (int256) {
          return a > b ? a : b;
      }
  
      /**
       * @dev Returns the smallest of two signed numbers.
       */
      function min(int256 a, int256 b) internal pure returns (int256) {
          return a < b ? a : b;
      }
  
      /**
       * @dev Returns the average of two signed numbers without overflow.
       * The result is rounded towards zero.
       */
      function average(int256 a, int256 b) internal pure returns (int256) {
          // Formula from the book "Hacker's Delight"
          int256 x = (a & b) + ((a ^ b) >> 1);
          return x + (int256(uint256(x) >> 255) & (a ^ b));
      }
  
      /**
       * @dev Returns the absolute unsigned value of a signed value.
       */
      function abs(int256 n) internal pure returns (uint256) {
          unchecked {
              // must be unchecked in order to support `n = type(int256).min`
              return uint256(n >= 0 ? n : -n);
          }
      }
  }
  
  // File: @openzeppelin/contracts/utils/math/Math.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (utils/math/Math.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Standard math utilities missing in the Solidity language.
   */
  library Math {
      enum Rounding {
          Down, // Toward negative infinity
          Up, // Toward infinity
          Zero // Toward zero
      }
  
      /**
       * @dev Returns the largest of two numbers.
       */
      function max(uint256 a, uint256 b) internal pure returns (uint256) {
          return a > b ? a : b;
      }
  
      /**
       * @dev Returns the smallest of two numbers.
       */
      function min(uint256 a, uint256 b) internal pure returns (uint256) {
          return a < b ? a : b;
      }
  
      /**
       * @dev Returns the average of two numbers. The result is rounded towards
       * zero.
       */
      function average(uint256 a, uint256 b) internal pure returns (uint256) {
          // (a + b) / 2 can overflow.
          return (a & b) + (a ^ b) / 2;
      }
  
      /**
       * @dev Returns the ceiling of the division of two numbers.
       *
       * This differs from standard division with `/` in that it rounds up instead
       * of rounding down.
       */
      function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
          // (a + b - 1) / b can overflow on addition, so we distribute.
          return a == 0 ? 0 : (a - 1) / b + 1;
      }
  
      /**
       * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
       * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
       * with further edits by Uniswap Labs also under MIT license.
       */
      function mulDiv(uint256 x, uint256 y, uint256 denominator) internal pure returns (uint256 result) {
          unchecked {
              // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
              // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
              // variables such that product = prod1 * 2^256 + prod0.
              uint256 prod0; // Least significant 256 bits of the product
              uint256 prod1; // Most significant 256 bits of the product
              assembly {
                  let mm := mulmod(x, y, not(0))
                  prod0 := mul(x, y)
                  prod1 := sub(sub(mm, prod0), lt(mm, prod0))
              }
  
              // Handle non-overflow cases, 256 by 256 division.
              if (prod1 == 0) {
                  // Solidity will revert if denominator == 0, unlike the div opcode on its own.
                  // The surrounding unchecked block does not change this fact.
                  // See https://docs.soliditylang.org/en/latest/control-structures.html#checked-or-unchecked-arithmetic.
                  return prod0 / denominator;
              }
  
              // Make sure the result is less than 2^256. Also prevents denominator == 0.
              require(denominator > prod1, "Math: mulDiv overflow");
  
              ///////////////////////////////////////////////
              // 512 by 256 division.
              ///////////////////////////////////////////////
  
              // Make division exact by subtracting the remainder from [prod1 prod0].
              uint256 remainder;
              assembly {
                  // Compute remainder using mulmod.
                  remainder := mulmod(x, y, denominator)
  
                  // Subtract 256 bit number from 512 bit number.
                  prod1 := sub(prod1, gt(remainder, prod0))
                  prod0 := sub(prod0, remainder)
              }
  
              // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
              // See https://cs.stackexchange.com/q/138556/92363.
  
              // Does not overflow because the denominator cannot be zero at this stage in the function.
              uint256 twos = denominator & (~denominator + 1);
              assembly {
                  // Divide denominator by twos.
                  denominator := div(denominator, twos)
  
                  // Divide [prod1 prod0] by twos.
                  prod0 := div(prod0, twos)
  
                  // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                  twos := add(div(sub(0, twos), twos), 1)
              }
  
              // Shift in bits from prod1 into prod0.
              prod0 |= prod1 * twos;
  
              // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
              // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
              // four bits. That is, denominator * inv = 1 mod 2^4.
              uint256 inverse = (3 * denominator) ^ 2;
  
              // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
              // in modular arithmetic, doubling the correct bits in each step.
              inverse *= 2 - denominator * inverse; // inverse mod 2^8
              inverse *= 2 - denominator * inverse; // inverse mod 2^16
              inverse *= 2 - denominator * inverse; // inverse mod 2^32
              inverse *= 2 - denominator * inverse; // inverse mod 2^64
              inverse *= 2 - denominator * inverse; // inverse mod 2^128
              inverse *= 2 - denominator * inverse; // inverse mod 2^256
  
              // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
              // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
              // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
              // is no longer required.
              result = prod0 * inverse;
              return result;
          }
      }
  
      /**
       * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
       */
      function mulDiv(uint256 x, uint256 y, uint256 denominator, Rounding rounding) internal pure returns (uint256) {
          uint256 result = mulDiv(x, y, denominator);
          if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
              result += 1;
          }
          return result;
      }
  
      /**
       * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
       *
       * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
       */
      function sqrt(uint256 a) internal pure returns (uint256) {
          if (a == 0) {
              return 0;
          }
  
          // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
          //
          // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
          // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
          //
          // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
          // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
          // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
          //
          // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
          uint256 result = 1 << (log2(a) >> 1);
  
          // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
          // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
          // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
          // into the expected uint128 result.
          unchecked {
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              result = (result + a / result) >> 1;
              return min(result, a / result);
          }
      }
  
      /**
       * @notice Calculates sqrt(a), following the selected rounding direction.
       */
      function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = sqrt(a);
              return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 2, rounded down, of a positive value.
       * Returns 0 if given 0.
       */
      function log2(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >> 128 > 0) {
                  value >>= 128;
                  result += 128;
              }
              if (value >> 64 > 0) {
                  value >>= 64;
                  result += 64;
              }
              if (value >> 32 > 0) {
                  value >>= 32;
                  result += 32;
              }
              if (value >> 16 > 0) {
                  value >>= 16;
                  result += 16;
              }
              if (value >> 8 > 0) {
                  value >>= 8;
                  result += 8;
              }
              if (value >> 4 > 0) {
                  value >>= 4;
                  result += 4;
              }
              if (value >> 2 > 0) {
                  value >>= 2;
                  result += 2;
              }
              if (value >> 1 > 0) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = log2(value);
              return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 10, rounded down, of a positive value.
       * Returns 0 if given 0.
       */
      function log10(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >= 10 ** 64) {
                  value /= 10 ** 64;
                  result += 64;
              }
              if (value >= 10 ** 32) {
                  value /= 10 ** 32;
                  result += 32;
              }
              if (value >= 10 ** 16) {
                  value /= 10 ** 16;
                  result += 16;
              }
              if (value >= 10 ** 8) {
                  value /= 10 ** 8;
                  result += 8;
              }
              if (value >= 10 ** 4) {
                  value /= 10 ** 4;
                  result += 4;
              }
              if (value >= 10 ** 2) {
                  value /= 10 ** 2;
                  result += 2;
              }
              if (value >= 10 ** 1) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = log10(value);
              return result + (rounding == Rounding.Up && 10 ** result < value ? 1 : 0);
          }
      }
  
      /**
       * @dev Return the log in base 256, rounded down, of a positive value.
       * Returns 0 if given 0.
       *
       * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
       */
      function log256(uint256 value) internal pure returns (uint256) {
          uint256 result = 0;
          unchecked {
              if (value >> 128 > 0) {
                  value >>= 128;
                  result += 16;
              }
              if (value >> 64 > 0) {
                  value >>= 64;
                  result += 8;
              }
              if (value >> 32 > 0) {
                  value >>= 32;
                  result += 4;
              }
              if (value >> 16 > 0) {
                  value >>= 16;
                  result += 2;
              }
              if (value >> 8 > 0) {
                  result += 1;
              }
          }
          return result;
      }
  
      /**
       * @dev Return the log in base 256, following the selected rounding direction, of a positive value.
       * Returns 0 if given 0.
       */
      function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
          unchecked {
              uint256 result = log256(value);
              return result + (rounding == Rounding.Up && 1 << (result << 3) < value ? 1 : 0);
          }
      }
  }
  
  // File: @openzeppelin/contracts/utils/Strings.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (utils/Strings.sol)
  
  pragma solidity ^0.8.0;
  
  
  
  /**
   * @dev String operations.
   */
  library Strings {
      bytes16 private constant _SYMBOLS = "0123456789abcdef";
      uint8 private constant _ADDRESS_LENGTH = 20;
  
      /**
       * @dev Converts a `uint256` to its ASCII `string` decimal representation.
       */
      function toString(uint256 value) internal pure returns (string memory) {
          unchecked {
              uint256 length = Math.log10(value) + 1;
              string memory buffer = new string(length);
              uint256 ptr;
              /// @solidity memory-safe-assembly
              assembly {
                  ptr := add(buffer, add(32, length))
              }
              while (true) {
                  ptr--;
                  /// @solidity memory-safe-assembly
                  assembly {
                      mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                  }
                  value /= 10;
                  if (value == 0) break;
              }
              return buffer;
          }
      }
  
      /**
       * @dev Converts a `int256` to its ASCII `string` decimal representation.
       */
      function toString(int256 value) internal pure returns (string memory) {
          return string(abi.encodePacked(value < 0 ? "-" : "", toString(SignedMath.abs(value))));
      }
  
      /**
       * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
       */
      function toHexString(uint256 value) internal pure returns (string memory) {
          unchecked {
              return toHexString(value, Math.log256(value) + 1);
          }
      }
  
      /**
       * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
       */
      function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
          bytes memory buffer = new bytes(2 * length + 2);
          buffer[0] = "0";
          buffer[1] = "x";
          for (uint256 i = 2 * length + 1; i > 1; --i) {
              buffer[i] = _SYMBOLS[value & 0xf];
              value >>= 4;
          }
          require(value == 0, "Strings: hex length insufficient");
          return string(buffer);
      }
  
      /**
       * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
       */
      function toHexString(address addr) internal pure returns (string memory) {
          return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
      }
  
      /**
       * @dev Returns true if the two strings are equal.
       */
      function equal(string memory a, string memory b) internal pure returns (bool) {
          return keccak256(bytes(a)) == keccak256(bytes(b));
      }
  }
  
  // File: @openzeppelin/contracts/utils/math/SafeMath.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (utils/math/SafeMath.sol)
  
  pragma solidity ^0.8.0;
  
  // CAUTION
  // This version of SafeMath should only be used with Solidity 0.8 or later,
  // because it relies on the compiler's built in overflow checks.
  
  /**
   * @dev Wrappers over Solidity's arithmetic operations.
   *
   * NOTE: `SafeMath` is generally not needed starting with Solidity 0.8, since the compiler
   * now has built in overflow checking.
   */
  library SafeMath {
      /**
       * @dev Returns the addition of two unsigned integers, with an overflow flag.
       *
       * _Available since v3.4._
       */
      function tryAdd(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              uint256 c = a + b;
              if (c < a) return (false, 0);
              return (true, c);
          }
      }
  
      /**
       * @dev Returns the subtraction of two unsigned integers, with an overflow flag.
       *
       * _Available since v3.4._
       */
      function trySub(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              if (b > a) return (false, 0);
              return (true, a - b);
          }
      }
  
      /**
       * @dev Returns the multiplication of two unsigned integers, with an overflow flag.
       *
       * _Available since v3.4._
       */
      function tryMul(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              // Gas optimization: this is cheaper than requiring 'a' not being zero, but the
              // benefit is lost if 'b' is also tested.
              // See: https://github.com/OpenZeppelin/openzeppelin-contracts/pull/522
              if (a == 0) return (true, 0);
              uint256 c = a * b;
              if (c / a != b) return (false, 0);
              return (true, c);
          }
      }
  
      /**
       * @dev Returns the division of two unsigned integers, with a division by zero flag.
       *
       * _Available since v3.4._
       */
      function tryDiv(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              if (b == 0) return (false, 0);
              return (true, a / b);
          }
      }
  
      /**
       * @dev Returns the remainder of dividing two unsigned integers, with a division by zero flag.
       *
       * _Available since v3.4._
       */
      function tryMod(uint256 a, uint256 b) internal pure returns (bool, uint256) {
          unchecked {
              if (b == 0) return (false, 0);
              return (true, a % b);
          }
      }
  
      /**
       * @dev Returns the addition of two unsigned integers, reverting on
       * overflow.
       *
       * Counterpart to Solidity's `+` operator.
       *
       * Requirements:
       *
       * - Addition cannot overflow.
       */
      function add(uint256 a, uint256 b) internal pure returns (uint256) {
          return a + b;
      }
  
      /**
       * @dev Returns the subtraction of two unsigned integers, reverting on
       * overflow (when the result is negative).
       *
       * Counterpart to Solidity's `-` operator.
       *
       * Requirements:
       *
       * - Subtraction cannot overflow.
       */
      function sub(uint256 a, uint256 b) internal pure returns (uint256) {
          return a - b;
      }
  
      /**
       * @dev Returns the multiplication of two unsigned integers, reverting on
       * overflow.
       *
       * Counterpart to Solidity's `*` operator.
       *
       * Requirements:
       *
       * - Multiplication cannot overflow.
       */
      function mul(uint256 a, uint256 b) internal pure returns (uint256) {
          return a * b;
      }
  
      /**
       * @dev Returns the integer division of two unsigned integers, reverting on
       * division by zero. The result is rounded towards zero.
       *
       * Counterpart to Solidity's `/` operator.
       *
       * Requirements:
       *
       * - The divisor cannot be zero.
       */
      function div(uint256 a, uint256 b) internal pure returns (uint256) {
          return a / b;
      }
  
      /**
       * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
       * reverting when dividing by zero.
       *
       * Counterpart to Solidity's `%` operator. This function uses a `revert`
       * opcode (which leaves remaining gas untouched) while Solidity uses an
       * invalid opcode to revert (consuming all remaining gas).
       *
       * Requirements:
       *
       * - The divisor cannot be zero.
       */
      function mod(uint256 a, uint256 b) internal pure returns (uint256) {
          return a % b;
      }
  
      /**
       * @dev Returns the subtraction of two unsigned integers, reverting with custom message on
       * overflow (when the result is negative).
       *
       * CAUTION: This function is deprecated because it requires allocating memory for the error
       * message unnecessarily. For custom revert reasons use {trySub}.
       *
       * Counterpart to Solidity's `-` operator.
       *
       * Requirements:
       *
       * - Subtraction cannot overflow.
       */
      function sub(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
          unchecked {
              require(b <= a, errorMessage);
              return a - b;
          }
      }
  
      /**
       * @dev Returns the integer division of two unsigned integers, reverting with custom message on
       * division by zero. The result is rounded towards zero.
       *
       * Counterpart to Solidity's `/` operator. Note: this function uses a
       * `revert` opcode (which leaves remaining gas untouched) while Solidity
       * uses an invalid opcode to revert (consuming all remaining gas).
       *
       * Requirements:
       *
       * - The divisor cannot be zero.
       */
      function div(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
          unchecked {
              require(b > 0, errorMessage);
              return a / b;
          }
      }
  
      /**
       * @dev Returns the remainder of dividing two unsigned integers. (unsigned integer modulo),
       * reverting with custom message when dividing by zero.
       *
       * CAUTION: This function is deprecated because it requires allocating memory for the error
       * message unnecessarily. For custom revert reasons use {tryMod}.
       *
       * Counterpart to Solidity's `%` operator. This function uses a `revert`
       * opcode (which leaves remaining gas untouched) while Solidity uses an
       * invalid opcode to revert (consuming all remaining gas).
       *
       * Requirements:
       *
       * - The divisor cannot be zero.
       */
      function mod(uint256 a, uint256 b, string memory errorMessage) internal pure returns (uint256) {
          unchecked {
              require(b > 0, errorMessage);
              return a % b;
          }
      }
  }
  
  // File: @openzeppelin/contracts/security/ReentrancyGuard.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Contract module that helps prevent reentrant calls to a function.
   *
   * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
   * available, which can be applied to functions to make sure there are no nested
   * (reentrant) calls to them.
   *
   * Note that because there is a single `nonReentrant` guard, functions marked as
   * `nonReentrant` may not call one another. This can be worked around by making
   * those functions `private`, and then adding `external` `nonReentrant` entry
   * points to them.
   *
   * TIP: If you would like to learn more about reentrancy and alternative ways
   * to protect against it, check out our blog post
   * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
   */
  abstract contract ReentrancyGuard {
      // Booleans are more expensive than uint256 or any type that takes up a full
      // word because each write operation emits an extra SLOAD to first read the
      // slot's contents, replace the bits taken up by the boolean, and then write
      // back. This is the compiler's defense against contract upgrades and
      // pointer aliasing, and it cannot be disabled.
  
      // The values being non-zero value makes deployment a bit more expensive,
      // but in exchange the refund on every call to nonReentrant will be lower in
      // amount. Since refunds are capped to a percentage of the total
      // transaction's gas, it is best to keep them low in cases like this one, to
      // increase the likelihood of the full refund coming into effect.
      uint256 private constant _NOT_ENTERED = 1;
      uint256 private constant _ENTERED = 2;
  
      uint256 private _status;
  
      constructor() {
          _status = _NOT_ENTERED;
      }
  
      /**
       * @dev Prevents a contract from calling itself, directly or indirectly.
       * Calling a `nonReentrant` function from another `nonReentrant`
       * function is not supported. It is possible to prevent this from happening
       * by making the `nonReentrant` function external, and making it call a
       * `private` function that does the actual work.
       */
      modifier nonReentrant() {
          _nonReentrantBefore();
          _;
          _nonReentrantAfter();
      }
  
      function _nonReentrantBefore() private {
          // On the first call to nonReentrant, _status will be _NOT_ENTERED
          require(_status != _ENTERED, "ReentrancyGuard: reentrant call");
  
          // Any calls to nonReentrant after this point will fail
          _status = _ENTERED;
      }
  
      function _nonReentrantAfter() private {
          // By storing the original value once again, a refund is triggered (see
          // https://eips.ethereum.org/EIPS/eip-2200)
          _status = _NOT_ENTERED;
      }
  
      /**
       * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
       * `nonReentrant` function in the call stack.
       */
      function _reentrancyGuardEntered() internal view returns (bool) {
          return _status == _ENTERED;
      }
  }
  
  // File: @openzeppelin/contracts/utils/Address.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (utils/Address.sol)
  
  pragma solidity ^0.8.1;
  
  /**
   * @dev Collection of functions related to the address type
   */
  library Address {
      /**
       * @dev Returns true if `account` is a contract.
       *
       * [IMPORTANT]
       * ====
       * It is unsafe to assume that an address for which this function returns
       * false is an externally-owned account (EOA) and not a contract.
       *
       * Among others, `isContract` will return false for the following
       * types of addresses:
       *
       *  - an externally-owned account
       *  - a contract in construction
       *  - an address where a contract will be created
       *  - an address where a contract lived, but was destroyed
       *
       * Furthermore, `isContract` will also return true if the target contract within
       * the same transaction is already scheduled for destruction by `SELFDESTRUCT`,
       * which only has an effect at the end of a transaction.
       * ====
       *
       * [IMPORTANT]
       * ====
       * You shouldn't rely on `isContract` to protect against flash loan attacks!
       *
       * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
       * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
       * constructor.
       * ====
       */
      function isContract(address account) internal view returns (bool) {
          // This method relies on extcodesize/address.code.length, which returns 0
          // for contracts in construction, since the code is only stored at the end
          // of the constructor execution.
  
          return account.code.length > 0;
      }
  
      /**
       * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
       * `recipient`, forwarding all available gas and reverting on errors.
       *
       * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
       * of certain opcodes, possibly making contracts go over the 2300 gas limit
       * imposed by `transfer`, making them unable to receive funds via
       * `transfer`. {sendValue} removes this limitation.
       *
       * https://consensys.net/diligence/blog/2019/09/stop-using-soliditys-transfer-now/[Learn more].
       *
       * IMPORTANT: because control is transferred to `recipient`, care must be
       * taken to not create reentrancy vulnerabilities. Consider using
       * {ReentrancyGuard} or the
       * https://solidity.readthedocs.io/en/v0.8.0/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
       */
      function sendValue(address payable recipient, uint256 amount) internal {
          require(address(this).balance >= amount, "Address: insufficient balance");
  
          (bool success, ) = recipient.call{value: amount}("");
          require(success, "Address: unable to send value, recipient may have reverted");
      }
  
      /**
       * @dev Performs a Solidity function call using a low level `call`. A
       * plain `call` is an unsafe replacement for a function call: use this
       * function instead.
       *
       * If `target` reverts with a revert reason, it is bubbled up by this
       * function (like regular Solidity function calls).
       *
       * Returns the raw returned data. To convert to the expected return value,
       * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
       *
       * Requirements:
       *
       * - `target` must be a contract.
       * - calling `target` with `data` must not revert.
       *
       * _Available since v3.1._
       */
      function functionCall(address target, bytes memory data) internal returns (bytes memory) {
          return functionCallWithValue(target, data, 0, "Address: low-level call failed");
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
       * `errorMessage` as a fallback revert reason when `target` reverts.
       *
       * _Available since v3.1._
       */
      function functionCall(
          address target,
          bytes memory data,
          string memory errorMessage
      ) internal returns (bytes memory) {
          return functionCallWithValue(target, data, 0, errorMessage);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but also transferring `value` wei to `target`.
       *
       * Requirements:
       *
       * - the calling contract must have an ETH balance of at least `value`.
       * - the called Solidity function must be `payable`.
       *
       * _Available since v3.1._
       */
      function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
          return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
      }
  
      /**
       * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
       * with `errorMessage` as a fallback revert reason when `target` reverts.
       *
       * _Available since v3.1._
       */
      function functionCallWithValue(
          address target,
          bytes memory data,
          uint256 value,
          string memory errorMessage
      ) internal returns (bytes memory) {
          require(address(this).balance >= value, "Address: insufficient balance for call");
          (bool success, bytes memory returndata) = target.call{value: value}(data);
          return verifyCallResultFromTarget(target, success, returndata, errorMessage);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but performing a static call.
       *
       * _Available since v3.3._
       */
      function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
          return functionStaticCall(target, data, "Address: low-level static call failed");
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
       * but performing a static call.
       *
       * _Available since v3.3._
       */
      function functionStaticCall(
          address target,
          bytes memory data,
          string memory errorMessage
      ) internal view returns (bytes memory) {
          (bool success, bytes memory returndata) = target.staticcall(data);
          return verifyCallResultFromTarget(target, success, returndata, errorMessage);
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
       * but performing a delegate call.
       *
       * _Available since v3.4._
       */
      function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
          return functionDelegateCall(target, data, "Address: low-level delegate call failed");
      }
  
      /**
       * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
       * but performing a delegate call.
       *
       * _Available since v3.4._
       */
      function functionDelegateCall(
          address target,
          bytes memory data,
          string memory errorMessage
      ) internal returns (bytes memory) {
          (bool success, bytes memory returndata) = target.delegatecall(data);
          return verifyCallResultFromTarget(target, success, returndata, errorMessage);
      }
  
      /**
       * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
       * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
       *
       * _Available since v4.8._
       */
      function verifyCallResultFromTarget(
          address target,
          bool success,
          bytes memory returndata,
          string memory errorMessage
      ) internal view returns (bytes memory) {
          if (success) {
              if (returndata.length == 0) {
                  // only check isContract if the call was successful and the return data is empty
                  // otherwise we already know that it was a contract
                  require(isContract(target), "Address: call to non-contract");
              }
              return returndata;
          } else {
              _revert(returndata, errorMessage);
          }
      }
  
      /**
       * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
       * revert reason or using the provided one.
       *
       * _Available since v4.3._
       */
      function verifyCallResult(
          bool success,
          bytes memory returndata,
          string memory errorMessage
      ) internal pure returns (bytes memory) {
          if (success) {
              return returndata;
          } else {
              _revert(returndata, errorMessage);
          }
      }
  
      function _revert(bytes memory returndata, string memory errorMessage) private pure {
          // Look for revert reason and bubble it up if present
          if (returndata.length > 0) {
              // The easiest way to bubble the revert reason is using memory via assembly
              /// @solidity memory-safe-assembly
              assembly {
                  let returndata_size := mload(returndata)
                  revert(add(32, returndata), returndata_size)
              }
          } else {
              revert(errorMessage);
          }
      }
  }
  
  // File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Permit.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/extensions/IERC20Permit.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
   * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
   *
   * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
   * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
   * need to send a transaction, and thus is not required to hold Ether at all.
   */
  interface IERC20Permit {
      /**
       * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
       * given ``owner``'s signed approval.
       *
       * IMPORTANT: The same issues {IERC20-approve} has related to transaction
       * ordering also apply here.
       *
       * Emits an {Approval} event.
       *
       * Requirements:
       *
       * - `spender` cannot be the zero address.
       * - `deadline` must be a timestamp in the future.
       * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
       * over the EIP712-formatted function arguments.
       * - the signature must use ``owner``'s current nonce (see {nonces}).
       *
       * For more information on the signature format, see the
       * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
       * section].
       */
      function permit(
          address owner,
          address spender,
          uint256 value,
          uint256 deadline,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) external;
  
      /**
       * @dev Returns the current nonce for `owner`. This value must be
       * included whenever a signature is generated for {permit}.
       *
       * Every successful call to {permit} increases ``owner``'s nonce by one. This
       * prevents a signature from being used multiple times.
       */
      function nonces(address owner) external view returns (uint256);
  
      /**
       * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
       */
      // solhint-disable-next-line func-name-mixedcase
      function DOMAIN_SEPARATOR() external view returns (bytes32);
  }
  
  // File: @openzeppelin/contracts/token/ERC20/IERC20.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Interface of the ERC20 standard as defined in the EIP.
   */
  interface IERC20 {
      /**
       * @dev Emitted when `value` tokens are moved from one account (`from`) to
       * another (`to`).
       *
       * Note that `value` may be zero.
       */
      event Transfer(address indexed from, address indexed to, uint256 value);
  
      /**
       * @dev Emitted when the allowance of a `spender` for an `owner` is set by
       * a call to {approve}. `value` is the new allowance.
       */
      event Approval(address indexed owner, address indexed spender, uint256 value);
  
      /**
       * @dev Returns the amount of tokens in existence.
       */
      function totalSupply() external view returns (uint256);
  
      /**
       * @dev Returns the amount of tokens owned by `account`.
       */
      function balanceOf(address account) external view returns (uint256);
  
      /**
       * @dev Moves `amount` tokens from the caller's account to `to`.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transfer(address to, uint256 amount) external returns (bool);
  
      /**
       * @dev Returns the remaining number of tokens that `spender` will be
       * allowed to spend on behalf of `owner` through {transferFrom}. This is
       * zero by default.
       *
       * This value changes when {approve} or {transferFrom} are called.
       */
      function allowance(address owner, address spender) external view returns (uint256);
  
      /**
       * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * IMPORTANT: Beware that changing an allowance with this method brings the risk
       * that someone may use both the old and the new allowance by unfortunate
       * transaction ordering. One possible solution to mitigate this race
       * condition is to first reduce the spender's allowance to 0 and set the
       * desired value afterwards:
       * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
       *
       * Emits an {Approval} event.
       */
      function approve(address spender, uint256 amount) external returns (bool);
  
      /**
       * @dev Moves `amount` tokens from `from` to `to` using the
       * allowance mechanism. `amount` is then deducted from the caller's
       * allowance.
       *
       * Returns a boolean value indicating whether the operation succeeded.
       *
       * Emits a {Transfer} event.
       */
      function transferFrom(address from, address to, uint256 amount) external returns (bool);
  }
  
  // File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.3) (token/ERC20/utils/SafeERC20.sol)
  
  pragma solidity ^0.8.0;
  
  
  
  
  /**
   * @title SafeERC20
   * @dev Wrappers around ERC20 operations that throw on failure (when the token
   * contract returns false). Tokens that return no value (and instead revert or
   * throw on failure) are also supported, non-reverting calls are assumed to be
   * successful.
   * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
   * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
   */
  library SafeERC20 {
      using Address for address;
  
      /**
       * @dev Transfer `value` amount of `token` from the calling contract to `to`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeTransfer(IERC20 token, address to, uint256 value) internal {
          _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
      }
  
      /**
       * @dev Transfer `value` amount of `token` from `from` to `to`, spending the approval given by `from` to the
       * calling contract. If `token` returns no value, non-reverting calls are assumed to be successful.
       */
      function safeTransferFrom(IERC20 token, address from, address to, uint256 value) internal {
          _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
      }
  
      /**
       * @dev Deprecated. This function has issues similar to the ones found in
       * {IERC20-approve}, and its usage is discouraged.
       *
       * Whenever possible, use {safeIncreaseAllowance} and
       * {safeDecreaseAllowance} instead.
       */
      function safeApprove(IERC20 token, address spender, uint256 value) internal {
          // safeApprove should only be called when setting an initial allowance,
          // or when resetting it to zero. To increase and decrease it, use
          // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
          require(
              (value == 0) || (token.allowance(address(this), spender) == 0),
              "SafeERC20: approve from non-zero to non-zero allowance"
          );
          _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
      }
  
      /**
       * @dev Increase the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeIncreaseAllowance(IERC20 token, address spender, uint256 value) internal {
          uint256 oldAllowance = token.allowance(address(this), spender);
          _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance + value));
      }
  
      /**
       * @dev Decrease the calling contract's allowance toward `spender` by `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful.
       */
      function safeDecreaseAllowance(IERC20 token, address spender, uint256 value) internal {
          unchecked {
              uint256 oldAllowance = token.allowance(address(this), spender);
              require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
              _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, oldAllowance - value));
          }
      }
  
      /**
       * @dev Set the calling contract's allowance toward `spender` to `value`. If `token` returns no value,
       * non-reverting calls are assumed to be successful. Meant to be used with tokens that require the approval
       * to be set to zero before setting it to a non-zero value, such as USDT.
       */
      function forceApprove(IERC20 token, address spender, uint256 value) internal {
          bytes memory approvalCall = abi.encodeWithSelector(token.approve.selector, spender, value);
  
          if (!_callOptionalReturnBool(token, approvalCall)) {
              _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, 0));
              _callOptionalReturn(token, approvalCall);
          }
      }
  
      /**
       * @dev Use a ERC-2612 signature to set the `owner` approval toward `spender` on `token`.
       * Revert on invalid signature.
       */
      function safePermit(
          IERC20Permit token,
          address owner,
          address spender,
          uint256 value,
          uint256 deadline,
          uint8 v,
          bytes32 r,
          bytes32 s
      ) internal {
          uint256 nonceBefore = token.nonces(owner);
          token.permit(owner, spender, value, deadline, v, r, s);
          uint256 nonceAfter = token.nonces(owner);
          require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
      }
  
      /**
       * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
       * on the return value: the return value is optional (but if data is returned, it must not be false).
       * @param token The token targeted by the call.
       * @param data The call data (encoded using abi.encode or one of its variants).
       */
      function _callOptionalReturn(IERC20 token, bytes memory data) private {
          // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
          // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
          // the target address contains contract code and also asserts for success in the low-level call.
  
          bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
          require(returndata.length == 0 || abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
      }
  
      /**
       * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
       * on the return value: the return value is optional (but if data is returned, it must not be false).
       * @param token The token targeted by the call.
       * @param data The call data (encoded using abi.encode or one of its variants).
       *
       * This is a variant of {_callOptionalReturn} that silents catches all reverts and returns a bool instead.
       */
      function _callOptionalReturnBool(IERC20 token, bytes memory data) private returns (bool) {
          // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
          // we're implementing it ourselves. We cannot use {Address-functionCall} here since this should return false
          // and not revert is the subcall reverts.
  
          (bool success, bytes memory returndata) = address(token).call(data);
          return
              success && (returndata.length == 0 || abi.decode(returndata, (bool))) && Address.isContract(address(token));
      }
  }
  
  // File: contracts/MCDStaking.sol
  
  
  pragma solidity ^0.8.18;
  
  
  
  
  
  
  
  
  interface NFT {
      function transferFrom(address sender, address receiver, uint256 tokenID) external;
  }
  
  contract MCDStaking is ERC721Holder, Ownable, ReentrancyGuard {
      
      using Strings for uint256;
      using SafeMath for uint256;
      using SafeERC20 for IERC20;
  
      address public immutable mcd = 0x13E0c7f86E28372D5FE7d9F002045e10eB288CF7;
      address public immutable treasury = 0xA682182447D462dc980a4d0CD0e81D0c4B6a235C;
  
      uint256 private immutable genesisMigrationDate = 1710777600;
      uint256 private constant genesisLockPeriod = 150 days;
  
      uint256 private constant mcdPercent = 70;
      uint256 private constant xmcdPercent = 30;
      uint256 private constant divider = 100;
  
      uint256 private constant timeStep = 1 days;
      uint256 private constant vestingPeriod = 90 days;
  
      uint256 public immutable stakingStart = 1711382400;
      uint256 public lastLiquidityRelease;
  
      uint256 public tokenEmission = 300000e18;
      uint256 public totalCollection = 2;
      uint256 public totalHashpower;
  
      struct StakeInfo {
          uint256 collectionID;
          address nftAddr;
          uint256 tokenID;
          uint256 stakeTime;
          uint256 hashpower;
          uint256 lastClaimTime;
          bool staked;
      }
  
      struct UserData {
          uint256 xMCD;
          uint256 hashpower;
      }
  
      mapping(address=>StakeInfo[]) public stakeInfo;
      mapping(address=>UserData) public userData;
      mapping(address=>bool) private admin;
      mapping(address=>uint256) public nftHashpower;
      mapping(uint256=>address) public nftContract;
      mapping(address=>uint256) public genesisMigratedToken;
      mapping(address=>mapping(uint256=>bool)) public stakedNFT;
  
      event Stake(address user, address nftAddr, uint256 tokenID, uint256 timestamp);
      event BatchStake(address user, address nftAddr, uint256[] tokenIDs, uint256 timestamp);
      event Unstake(address user, address nftAddr, uint256 tokenID, uint256 timestamp);
      event ClaimSingleNFTYield(address user, address nftAddr, uint256 tokenID, uint256 timestamp, uint256 totalYield);
      event ClaimAll(address user, uint256 timestamp, uint256 mcdEarned, uint256 xmcdEarned);
      event TransferXMCD(address sender, address receiver, uint256 transferAmount, uint256 timestamp);
      event DebitXMCD(address spender, uint256 spendAmount, uint256 timestamp);
      event CreditXMCD(address adminAddr, address receiver, uint256 creditAmount, uint256 timestamp);
      event GenesisStakeMigrate(address adminAddr, address holderWallet, uint256 tokenID, uint256 startTime, uint256 lastClaim);
      event GenesisStakeBatchMigrate(address adminAddr, address[] wallet, uint256[] tokenIDs, uint256 startTime, uint256 lastClaim);
      event GenesisTokenMigrate(address adminAddr, address receiver, uint256 amount);
      event GenesisTokenBatchMigrate(address adminAddr, address[] receiver, uint256[] amount);
      event GenesisTokenClaim(address user, uint256 mcdEarned, uint256 xmcdEarned, uint256 timestamp);
      event ReleaseToTreasury(address caller, uint256 unreleasedDays, uint256 releasedAmount, uint256 timestamp);
      event SetTokenEmission(address adminAddr, uint256 newEmission, uint256 timestamp);
      event AddNFTContract(address adminAddr, uint256 collectionID, address nftAddr, uint256 timestamp);
      event ManageAdmin(address newAdminAddr, bool state, uint256 timestamp);
      event UpdateNFTHashpower(address adminAddr, uint256 oldPower, uint256 newPower, uint256 timestamp);
      event SetGenesisContractAddr(address adminAddr, address genesisAddr, uint256 timestamp);
  
      constructor(
      ) {
          lastLiquidityRelease = block.timestamp;
          nftContract[1] = 0xeB248E38eC67DbE9Cc3DC82bB5eD6d1681939CD8;
          nftHashpower[nftContract[1]] = 20;
      }
  
      function stake(uint256 _collectionID, uint256 _tokenID) external nonReentrant {
          address nftAddr = nftContract[_collectionID];
          require(nftAddr != address(0), "Cannot be null address.");
          require(!stakedNFT[nftAddr][_tokenID], "This NFT is already staked.");
          stakeInfo[msg.sender].push(StakeInfo(_collectionID,nftAddr,_tokenID,block.timestamp,nftHashpower[nftAddr],block.timestamp,true));
          userData[msg.sender].hashpower += nftHashpower[nftAddr];
          totalHashpower += nftHashpower[nftAddr];
          stakedNFT[nftAddr][_tokenID] = true;
          NFT(nftAddr).transferFrom(msg.sender, address(this), _tokenID);
          emit Stake(msg.sender, nftAddr, _tokenID, block.timestamp);
      }
  
      function batchStake(uint256 _collectionID, uint256[] memory _tokenIDs) external nonReentrant {
          address nftAddr = nftContract[_collectionID];
          require(nftAddr != address(0), "Cannot be null address.");
          uint256 hashPowerIncrease = nftHashpower[nftAddr]*_tokenIDs.length;
          userData[msg.sender].hashpower += hashPowerIncrease;
          totalHashpower += hashPowerIncrease;
          uint256 tokensCount = _tokenIDs.length;
          for (uint256 i; i < tokensCount;) {
              require(!stakedNFT[nftAddr][_tokenIDs[i]], "One of the NFTs is already staked.");
              stakeInfo[msg.sender].push(StakeInfo(_collectionID,nftAddr,_tokenIDs[i],block.timestamp,nftHashpower[nftAddr],block.timestamp,true));
              stakedNFT[nftAddr][_tokenIDs[i]] = true;
              NFT(nftAddr).transferFrom(msg.sender, address(this), _tokenIDs[i]);
              unchecked {
                  ++i;
              }
          }
          emit BatchStake(msg.sender, nftAddr, _tokenIDs, block.timestamp);
      }
  
      function unstake(uint256 _collectionID, uint256 _tokenID) external nonReentrant {
          uint256 hashpowerDecrease;
          uint256 stakingRecords = stakeInfo[msg.sender].length;
          address nftAddr = nftContract[_collectionID];
          for (uint256 i; i < stakingRecords;) {
              if (stakeInfo[msg.sender][i].tokenID == _tokenID && stakeInfo[msg.sender][i].collectionID == _collectionID && stakeInfo[msg.sender][i].staked) {
                  uint256 mcdMined = getTokenYield(msg.sender, _collectionID, _tokenID);
                  if (block.timestamp >= stakeInfo[msg.sender][i].stakeTime + vestingPeriod) {
                      uint256 earnedMCD = mcdMined * mcdPercent / divider;
                      uint256 earnedXMCD = mcdMined - earnedMCD;
                      userData[msg.sender].xMCD += earnedXMCD;
                      IERC20(mcd).safeTransfer(msg.sender, earnedMCD);
                  } else {
                      IERC20(mcd).safeTransfer(treasury, mcdMined);
                  }
                  stakeInfo[msg.sender][i].lastClaimTime = block.timestamp;
                  stakeInfo[msg.sender][i].staked = false;
                  userData[msg.sender].hashpower -= stakeInfo[msg.sender][i].hashpower;
                  hashpowerDecrease += stakeInfo[msg.sender][i].hashpower;
                  stakedNFT[nftAddr][_tokenID] = false;
                  NFT(nftAddr).transferFrom(address(this), msg.sender, _tokenID);
                  break;
              }
              unchecked {
                  ++i;
              }
          }
          totalHashpower -= hashpowerDecrease;
          emit Unstake(msg.sender, nftAddr, _tokenID, block.timestamp);
      }
  
      function claimSingleNFTYield(uint256 _collectionID, uint256 _tokenID) external nonReentrant {
          uint256 stakingRecords = stakeInfo[msg.sender].length;
          for (uint256 i; i < stakingRecords;) {
              if (stakeInfo[msg.sender][i].tokenID == _tokenID && stakeInfo[msg.sender][i].collectionID == _collectionID && stakeInfo[msg.sender][i].staked) {
                  require(block.timestamp >= stakeInfo[msg.sender][i].lastClaimTime + timeStep, "Not yet 1 day since last claim.");
                  if (block.timestamp >= stakeInfo[msg.sender][i].stakeTime + vestingPeriod) {
                      uint256 mcdMined = getTokenYield(msg.sender, _collectionID, _tokenID);
                      require(mcdMined > 0, "Nothing to be claimed.");
                      uint256 earnedMCD = mcdMined * mcdPercent / divider;
                      uint256 earnedXMCD = mcdMined - earnedMCD;
                      userData[msg.sender].xMCD += earnedXMCD;
                      stakeInfo[msg.sender][i].lastClaimTime = block.timestamp;
                      IERC20(mcd).safeTransfer(msg.sender, earnedMCD);
                  }
                  break;
              }
              unchecked {
                  ++i;
              }
          }
          emit ClaimSingleNFTYield(msg.sender, nftContract[_collectionID], _tokenID, block.timestamp, getTokenYield(msg.sender, _collectionID, _tokenID));
      }
  
      function claimAll() external nonReentrant {
          uint256 totalYield;
          uint256 stakingRecords = stakeInfo[msg.sender].length;
          for (uint256 i; i < stakingRecords;) {
              if (stakeInfo[msg.sender][i].staked && block.timestamp >= stakeInfo[msg.sender][i].stakeTime + vestingPeriod) {
                  if (block.timestamp >= stakeInfo[msg.sender][i].lastClaimTime + timeStep) {
                      uint256 mcdMined = getTokenYield(msg.sender, stakeInfo[msg.sender][i].collectionID, stakeInfo[msg.sender][i].tokenID);
                      totalYield += mcdMined;
                      stakeInfo[msg.sender][i].lastClaimTime = block.timestamp;
                  }
              }
              unchecked {
                  ++i;
              }
          }
          require(totalYield > 0, "Nothing to be claimed.");
          uint256 earnedMCD = totalYield * mcdPercent / divider;
          uint256 earnedXMCD = totalYield - earnedMCD;
          userData[msg.sender].xMCD += earnedXMCD;
          IERC20(mcd).safeTransfer(msg.sender, earnedMCD);
          emit ClaimAll(msg.sender, block.timestamp, earnedMCD, earnedXMCD);
      }
  
      function getTokenYield(address _wallet, uint256 _collectionID, uint256 _tokenID) public view returns (uint256) {
          uint256 yield;
          uint256 stakingRecords = stakeInfo[_wallet].length;
          for (uint256 i; i < stakingRecords;) {
              if (stakeInfo[_wallet][i].tokenID == _tokenID && stakeInfo[_wallet][i].collectionID == _collectionID && stakeInfo[_wallet][i].staked) {
                  uint256 stakedPeriod = block.timestamp - stakeInfo[_wallet][i].lastClaimTime;
                  uint256 fullDayCount = stakedPeriod/timeStep;
                  yield = (tokenEmission*stakeInfo[_wallet][i].hashpower*fullDayCount/totalHashpower/30);
                  break;
              }
              if (i >= 20000) break;
              unchecked {
                  ++i;
              }
          }
          return yield;
      }
  
      function getTotalTokenYield(address _wallet) public view returns (uint256) {
          uint256 totalYield = 0;
          uint256 stakingRecords = stakeInfo[_wallet].length;
          for (uint256 i; i < stakingRecords;) {
              if (stakeInfo[_wallet][i].staked) {
                  uint256 stakedPeriod = block.timestamp - stakeInfo[_wallet][i].lastClaimTime;
                  uint256 fullDayCount = stakedPeriod/timeStep;
                  totalYield += (tokenEmission*stakeInfo[_wallet][i].hashpower*fullDayCount/totalHashpower/30);
              }
              if (i >= 20000) break;
              unchecked {
                  ++i;
              }
          }
          return totalYield;
      }
  
      function getTotalClaimable(address _wallet) public view returns (uint256) {
          uint256 totalClaimable = 0;
          uint256 stakingRecords = stakeInfo[_wallet].length;
          for (uint256 i; i < stakingRecords;) {
              if (stakeInfo[_wallet][i].staked && block.timestamp >= stakeInfo[msg.sender][i].stakeTime + vestingPeriod) {
                  uint256 stakedPeriod = block.timestamp - stakeInfo[_wallet][i].lastClaimTime;
                  uint256 fullDayCount = stakedPeriod/timeStep;
                  totalClaimable += (tokenEmission*stakeInfo[_wallet][i].hashpower*fullDayCount/totalHashpower/30);
              }
              if (i >= 20000) break;
              unchecked {
                  ++i;
              }
          }
          return totalClaimable;
      }
  
      function getCurDay() public view returns (uint256) {
          return (block.timestamp - stakingStart) / timeStep;
      }
  
      function transferXMCD(address _receiver, uint256 _transferAmount) external {
          require(_receiver != address(0), "Receiver cannot be null address.");
          require(userData[msg.sender].xMCD >= _transferAmount, "User insufficient xMCD balance.");
          userData[msg.sender].xMCD -= _transferAmount;
          userData[_receiver].xMCD += _transferAmount;
          emit TransferXMCD(msg.sender, _receiver, _transferAmount, block.timestamp);
      }
  
      function debitXMCD(address _spender, uint256 _debitAmount) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admin or owner can execute.");
          require(_spender != address(0), "Spender cannot be null address.");
          require(_debitAmount > 0, "Debit amount cannot be zero.");
          require(userData[_spender].xMCD >= _debitAmount, "User insufficient xMCD balance.");
          userData[_spender].xMCD -= _debitAmount;
          emit DebitXMCD(msg.sender, _debitAmount, block.timestamp);
      }
  
      function creditXMCD(address _receiver, uint256 _creditAmount) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admin or owner can execute.");
          require(_receiver != address(0), "Receiver cannot be null address.");
          require(_creditAmount > 0, "Credit amount cannot be zero.");
          userData[_receiver].xMCD += _creditAmount;
          emit CreditXMCD(msg.sender, _receiver, _creditAmount, block.timestamp);
      }
  
      function genesisStakeMigrate(address _wallet, uint256 _tokenID, uint256 _startTime, uint256 _lastClaim) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admin or owner can execute.");
          require(_wallet != address(0), "Cannot migrate to null address.");
          require(!stakedNFT[nftContract[0]][_tokenID], "This NFT is already staked.");
          require(_startTime > 0 && _lastClaim > 0, "Timestamp cannot be zero.");
          require(_startTime >= stakingStart, "Timestamp must be greater than staking start time.");
          require(_lastClaim >= _startTime, "Invalid last claim timestamp.");
          stakeInfo[_wallet].push(StakeInfo(0,nftContract[0],_tokenID,_startTime,nftHashpower[nftContract[0]],_lastClaim,true));
          userData[_wallet].hashpower += nftHashpower[nftContract[0]];
          totalHashpower += nftHashpower[nftContract[0]];
          stakedNFT[nftContract[0]][_tokenID] = true;
          emit GenesisStakeMigrate(msg.sender, _wallet, _tokenID, _startTime, _lastClaim);
      }
  
      function genesisStakeBatchMigrate(address[] memory _wallet, uint256[] memory _tokenIDs, uint256 _startTime, uint256 _lastClaim) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admin or owner can execute.");
          require(_wallet.length == _tokenIDs.length, "Unmatched data for wallet and token IDs.");
          require(_startTime > 0 && _lastClaim > 0, "Timestamp cannot be zero.");
          require(_startTime >= stakingStart, "Timestamp must be greater than staking start time.");
          require(_lastClaim >= _startTime, "Invalid last claim timestamp.");
          uint256 hashpowerIncreaseCount;
          uint256 walletCount = _wallet.length;
          for (uint256 i; i < walletCount;) {
              if (_wallet[i] != address(0)) {
                  require(!stakedNFT[nftContract[0]][_tokenIDs[i]], "One of the NFTs is already staked.");
                  stakeInfo[_wallet[i]].push(StakeInfo(0,nftContract[0],_tokenIDs[i],_startTime,nftHashpower[nftContract[0]],_lastClaim,true));
                  userData[_wallet[i]].hashpower += nftHashpower[nftContract[0]];
                  hashpowerIncreaseCount++;
                  stakedNFT[nftContract[0]][_tokenIDs[i]] = true;
              }
              unchecked {
                  ++i;
              }
          }
          totalHashpower += (nftHashpower[nftContract[0]]*hashpowerIncreaseCount);
          emit GenesisStakeBatchMigrate(msg.sender, _wallet, _tokenIDs, _startTime, _lastClaim);
      }
  
      function genesisTokenMigrate(address _wallet, uint256 _amount) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admin or owner can execute.");
          require(_wallet != address(0), "Cannot migrate to null address.");
          genesisMigratedToken[_wallet] = _amount;
          emit GenesisTokenMigrate(msg.sender, _wallet, _amount);
      }
  
      function genesisTokenBatchMigrate(address[] memory _wallet, uint256[] memory _amount) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admin or owner can execute.");
          require(_wallet.length == _amount.length, "Unmatched data for wallet and amount.");
          uint256 walletCount = _wallet.length;
          for (uint256 i; i < walletCount;) {
              if (_wallet[i] != address(0)) {
                  genesisMigratedToken[_wallet[i]] = _amount[i];
              }
              unchecked {
                  ++i;
              }
          }
          emit GenesisTokenBatchMigrate(msg.sender, _wallet, _amount);
      }
  
      function genesisTokenClaim() external nonReentrant {
          require(genesisMigratedToken[msg.sender] > 0, "No migrated token can be claimed.");
          require(block.timestamp >= genesisMigrationDate + genesisLockPeriod, "Migrated tokens still under locking period.");
          uint256 unlockedMCD = genesisMigratedToken[msg.sender] * mcdPercent / divider;
          uint256 earnedXMCD = genesisMigratedToken[msg.sender] - unlockedMCD;
          userData[msg.sender].xMCD += earnedXMCD;
          genesisMigratedToken[msg.sender] = 0;
          IERC20(mcd).safeTransfer(msg.sender, unlockedMCD);
          emit GenesisTokenClaim(msg.sender, unlockedMCD, earnedXMCD, block.timestamp);
      }
  
      function releaseToTreasury() external nonReentrant {
          require(block.timestamp >= lastLiquidityRelease + timeStep, "Not yet 24 hours since last released.");
          require(totalHashpower > 0, "No NFT is staked for mining.");
          uint256 releasableDays = (block.timestamp-lastLiquidityRelease)/timeStep;
          uint256 releasableMCD = tokenEmission*releasableDays*xmcdPercent/divider/30;
          lastLiquidityRelease += (releasableDays*timeStep);
          IERC20(mcd).safeTransfer(treasury, releasableMCD);
          emit ReleaseToTreasury(msg.sender, releasableDays, releasableMCD, block.timestamp);
      }
  
      function setTokenEmission(uint256 _newEmission) external {
          require(admin[msg.sender] || msg.sender == owner(), "Only admins or owner can execute.");
          tokenEmission = _newEmission;
          emit SetTokenEmission(msg.sender, _newEmission, block.timestamp);
      }
  
      function addNFTContract(address _contract, uint256 _power) external onlyOwner {
          require(_contract != address(0), "Cannot add null address.");
          require(_power > 0, "Power cannot be zero.");
          nftContract[totalCollection] = _contract;
          nftHashpower[_contract] = _power;
          totalCollection++;
          emit AddNFTContract(msg.sender, totalCollection, _contract, block.timestamp);
      }
  
      function manageAdmin(address _adminAddr, bool _state) external onlyOwner {
          require(_adminAddr != address(0), "Cannot add null address.");
          admin[_adminAddr] = _state;
          emit ManageAdmin(_adminAddr, _state, block.timestamp);
      }
  
      function updateNFTHashpower(address _nftContract, uint256 _power) external onlyOwner {
          require(_nftContract != address(0), "Invalid contract address.");
          uint256 oldPower = nftHashpower[_nftContract];
          nftHashpower[_nftContract] = _power;
          emit UpdateNFTHashpower(msg.sender, oldPower, _power, block.timestamp);
      }
  
      function setGenesisContractAddr(address _genesisContract) external onlyOwner {
          require(nftContract[0] == address(0), "Genesis contract already set.");
          nftContract[0] = _genesisContract;
          nftHashpower[nftContract[0]] = 100;
          emit SetGenesisContractAddr(msg.sender, _genesisContract, block.timestamp);
      }
  
      function getUserStakingInfo(address _userAddr) external view returns(StakeInfo[] memory) {
          return stakeInfo[_userAddr];
      }
  
      function getUserData(address _userAddr) external view returns(UserData memory) {
          return userData[_userAddr];
      }
  
  }